FirstOrderSolve - A MAPLE Package

Sebastian Falkensteiner*
joint work with J. Cano, J.R. Sendra and F. Boulier, D. Robertz

Research Institute for Symbolic Computation (RISC)
Johannes Kepler University Linz, Austria

Tagung der Fachgruppe Computeralgebra,
March 9th, 2022

* The speaker is partly supported by the Austrian Science Fund (FWF): P31327-N32 and by the grant PID2020-113192GB-I00 (Mathematical Visualization: Foundations, Algorithms and Applications) from the Spanish MICINN.

References

[1] "*Existence and Convergence of Puiseux Series Solutions for First Order Autonomous Differential Equations*", J. Cano, S. Falkensteiner, J.R. Sendra, Journal of Symbolic Computation, 2020.
[2] "*Puiseux Series and Algebraic Solutions of First Order Autonomous AODEs - A MAPLE Package*", F. Boulier, J. Cano, S. Falkensteiner, J.R. Sendra, Communications in Computer and Information Science, 2021.
[3] "*Algebraic and Puiseux series solutions of systems of autonomous algebraic ODEs of dimension one in several variables*", J. Cano, S. Falkensteiner, D. Robertz, J.R. Sendra, Journal of Symbolic Computation, 2022.

Terminology

AODE ... Algebraic Ordinary Differential Equation
Autonomous / Constant Coefficients ... independent variable does not explicitly occur
Order ... Highest derivative occuring in the differential equation

An example of an autonomous AODE of order 1:

```
> F1 := ( (y'(x) - 1)^2 + y(x)^2)^3 - 4·(y'(x) - 1)^2·y(x)^2;
```

$$F1 := \left(\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) - 1\right)^2 + y(x)^2\right)^3 - 4\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) - 1\right)^2 y(x)^2 \tag{3.1.1}$$

Manipulations on AODEs are supported by the package **'DifferentialAlgebra'**.
For example we can compute the **initial** and **separant** of a differential polynomial.

```
> DifferentialAlgebra:-Tools:-Initial(F1, DifferentialAlgebra:-
      DifferentialRing(derivations = [x], blocks = [y]));
  DifferentialAlgebra:-Tools:-Separant(F1, DifferentialAlgebra:-
      DifferentialRing(derivations = [x], blocks = [y]));
```

$$1$$

$$6\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^5 - 30\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^4 + 12\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^3 y(x)^2 + 60\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^3 - 36\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^2 y(x)^2 - 60\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right)^2 + 6\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right) y(x)^4 + 28\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right) y(x)^2 + 30\,\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) - 6\, y(x)^4 - 4\, y(x)^2 - 6 \tag{3.1.2}$$

Puiseux Series ... Power Series in a variable x expanded around some x0 with fractional exponents such that the denominator is bounded and there are only finitely many terms with negative exponents (In the case where x0=infinity: finitely many positive exponents)
Ramification Index ... the least common multiple of the denominator of the exponents of a formal Puiseux Series

Puiseux series and its manipulations are already built-in. For formal power series with non-negative integer exponents computations can alternatively be achieved by using the package '**powseries**'.
Some examples of Puiseux Series expanded around 0 or infinity, respectively:

> $series(\mathrm{sqrt}(\sin(x)), x=0, 4);$

$$\sqrt{x} - \frac{x^{5/2}}{12} + \mathrm{O}(x^{9/2}) \qquad \textbf{(3.2.1)}$$

> $series\left(\frac{1}{1-x-\exp(x)}, x=0, 5\right);$

$$-\frac{1}{2}x^{-1} + \frac{1}{8} + \frac{1}{96}x - \frac{1}{384}x^2 + \mathrm{O}(x^3) \qquad \textbf{(3.2.2)}$$

> $series\left(\frac{1}{1+x}, x=\mathrm{infinity}, 5\right);$

$$\frac{1}{x} - \frac{1}{x^2} + \frac{1}{x^3} - \frac{1}{x^4} + \mathrm{O}\left(\frac{1}{x^5}\right) \qquad \textbf{(3.2.3)}$$

Goal and Methodology

Goal:
We are interested in Puisuex Series solutions of (systems of) autonomous AODEs.
Since we handle autonomous differential equations, we can shift the expansion point and only two cases have to be considered: 0 and infinity.

Methodology:
As described in the references, we are following an algebraic geometric approach:
For now let us consider a single autonomous AODE of order 1, given by F=0 with F $\in$ [y,y'], and solutions expanded around 0.

1) Compute a local parametrization A(T) of the implicitly defined curve

$$C(F) = \{(a,b) \in (\ \cup \{\ \})^2 \mid F(a,b)=0\}$$

centered at the curve-point (y0,p0) $\in$ C(F), where y(0)=y0 is the initial value of the possible solutions, by using '**algcurves:-puiseux**'.

> $G1 := subs(\{y'(x) = z, y(x) = y\}, F1);$

$$G1 := \left((z-1)^2 + y^2\right)^3 - 4\,(z-1)^2\,y^2 \qquad \textbf{(4.2.1)}$$

> $algcurves\text{:-}plot\_real\_curve(G1, y, z);$



> $ASet := algcurves\text{:-}puiseux(G1, y=0, z, 3, T);$
> $A := [rhs(ASet[3][1]), rhs(ASet[3][2])]:$

$$ASet := \left\{\left[y=T, z=1-\frac{T^2}{2}\right], \left[y=T, z=1+\frac{T^2}{2}\right], \left[y=-2\,T^2, z=1+\frac{15}{16}\,T^5+\frac{3}{2}\,T^3 - 2\,T\right], \left[y=2\,T^2, z=1-\frac{15}{16}\,T^5-\frac{3}{2}\,T^3+2\,T\right]\right\} \quad \textbf{(4.3.1)}$$

> $plot([op(A), T=-0.5\,..0.5], scaling=constrained);$



2) A curve branch is defined by an **equivalence class** of a local parametrization with the **substitution of formal power series of order 1**.
The equivalence class is also called a **place** of the curve C(F).

> $B := subs\left(\left\{T=-T^3+2\cdot T^2+2\cdot T\right\}, A\right);$

$$B := \left[-2\left(-T^3+2\,T^2+2\,T\right)^2, 1+\frac{15\left(-T^3+2\,T^2+2\,T\right)^5}{16}+\frac{3\left(-T^3+2\,T^2+2\,T\right)^3}{2}+2\,T^3-4\,T^2-4\,T\right] \quad \textbf{(4.4.1)}$$

> $plot([[op(A), T=-0.5\,..0.5], [op(B), T=-0.5\,..0.2]], scaling=constrained);$



Goal and Methodology

Lemma: A (non-constant) solution y(x) of F(y,y')=0 defines a local parametrization by **(y(x^n),y'(x^n))** of C(F), where n is the ramification index.
We call (y(x^n),y'(x^n)) a **solution parametrization** of C(F).

Goal (reformulation): Find the places containing solution parametrizations and find the solution parametrizations in those places.

The orders of the local parametrizations in the same place are equal. Since the order ord (y'(x^n)) is equal to n-1+ord(y(x^n)), the following criterion is a necessary condition for the existence of a parametrization corresponding to a solution.

3) Check whether

$$\mathbf{n = ord(\frac{d}{dT}A[1]) - ord(A[2])+1 > 0}.$$

$$> \quad n := ldegree\left(\frac{d}{dT}(A[1])\right) - ldegree(A[2]) + 1;$$

$$n := 2 \qquad \textbf{(5.1.1)}$$

If the necessary condition is fulfilled for a place given by the local parametrization A(t)=(a1(t),a2(t)), we are looking for a reparametrization A(s(t)) which is a solution parametrization. More precisely, we solve the **associated differential equation** arising from (y(x^n),y'(x^n))=(a1(s(t)),a2(s(t))) where s(t) ∈ [[t]] is of order one.

4) Solve

a1'(s(t)) s'(t) = n t^(n-1) a2(s(t))

for s(t).

> $S := add\left(seq\left(s[i]\cdot t^i, i=1..6\right)\right);$

$$S := s_6 t^6 + s_5 t^5 + s_4 t^4 + s_3 t^3 + s_2 t^2 + s_1 t \quad \textbf{(5.2.1)}$$

> $coeffexpr := subs\left(\{T=S\}, \frac{d}{dT}(A[1]) \cdot \frac{d}{dt}(S) - n \cdot t^{n-1} \cdot A[2]\right):$

> $coeffexpr := [seq(coeff(coeffexpr, t, i) = 0, i=1..6)];$

$$coeffexpr := \left[-4 s_1^2 - 2 = 0, -12 s_1 s_2 + 4 s_1 = 0, -16 s_1 s_3 - 8 s_2^2 + 4 s_2 = 0, -3 s_1^3 - 20 s_1 s_4 - 20 s_2 s_3 + 4 s_3 = 0, -9 s_1^2 s_2 - 24 s_1 s_5 - 24 s_2 s_4 - 12 s_3^2 + 4 s_4 = 0, -28 s_1 s_6 - 28 s_2 s_5 - 28 s_3 s_4 - \frac{15 s_1^5}{8} - 3 s_1 \left(2 s_1 s_3 + s_2^2\right) - 6 s_2^2 s_1 - 3 s_3 s_1^2 + 4 s_5 = 0\right] \quad \textbf{(5.3.1)}$$

The first equation is of degree n in s1. The following equations are affine linear in the leading variable.
In particular, the **existence** of the solution is guaranteed and, after choosing s1, the solution is **unique**.

> $reparaList := [allvalues(op(solve(coeffexpr, [seq(s[i], i=1..6)])))];$

$$reparaList := \left[\left[s_1 = -\frac{I}{2}\sqrt{2}, s_2 = \frac{1}{3}, s_3 = \frac{I}{36}\sqrt{2}, s_4 = \frac{89}{1080}, s_5 = \frac{107\, I}{2160}\sqrt{2}, s_6 = -\frac{41297}{1088640}\right], \left[s_1 = \frac{I}{2}\sqrt{2}, s_2 = \frac{1}{3}, s_3 = -\frac{I}{36}\sqrt{2}, s_4 = \frac{89}{1080}, s_5 = -\frac{107\, I}{2160}\sqrt{2}, s_6 = -\frac{41297}{1088640}\right]\right] \quad \textbf{(5.3.2)}$$

5) The expressions

a1(s(x^(1/n)))

are truncations of the solution.

```
> solTruncation := subs({T=subs(reparaList[1], S)}, A[1]) :
  solTruncation := subs({t=x^(1/n)}, solTruncation) :
  solTruncation := simplify(expand(solTruncation));
```

$$solTruncation := x + \frac{32663\,x^4}{680400} - \frac{x^2}{3} - \frac{247\,x^3}{810} - \frac{1705442209\,x^6}{592568524800} + \frac{6560581\,x^5}{293932800} + \frac{4418779\,\mathrm{I}\,x^{11/2}\sqrt{2}}{587865600} - \frac{593447\,\mathrm{I}\,x^{9/2}\sqrt{2}}{48988800} - \frac{9137\,\mathrm{I}\,x^{7/2}\sqrt{2}}{60480} + \frac{23\,\mathrm{I}\,x^{5/2}\sqrt{2}}{180} + \frac{2\,\mathrm{I}\sqrt{2}\,x^{3/2}}{3} \quad (5.4.1)$$

```
> solCheck := expand(eval(subs(y(x)=solTruncation, F1))) :
  seq(subs({x=0}, diff(solCheck, [x$i])), i=0..5);
```

$$0, 0, 0, 0, 0, 0 \quad (5.4.2)$$

We computed a solution with initial value y(0)=0, where the curve defined by G1 has a singularity.
For almost all curve points (those which do not have a vertical tangent) it is possible to use the **Implicit Function Theorem**. By handling these particular points as we do, we obtain 'all' solutions of the given differential equation.

Results

From this procedure we obtain the following results (see [1]).

Theorem 1: All formal Puiseux series solutions of an AODE F(y,y')=0 are convergent.

Theorem 2: For a finite expansion point, there is an effective bound N (depending only on deg(F;y), deg(F;y')) such that a truncated solution modulo x^N can extended uniquely to a Puiseux series solution of F(y,y')=0.

In the case of the expansion point inifnity, Theorem 2 holds except for the uniqueness. The reason is an arbitrary constant which appears in the computations.

FirstOrderSolve

The algorithm indicated above is implemented in the software-package FirstOrderSolve available at https://risc.jku.at/sw/firstordersolve/

```
> restart;
> F1 := ((y'(x)-1)^2+y(x)^2)^3 - 4*(y'(x)-1)^2*y(x)^2:
> libname := libname,
      "C:/Users/SWE/Documents/Research/Maple Projects/Autonomous AODEs":
> with(FirstOrderSolve);
```

$$[AlgebraicSolution, GenericSolutionTruncation, ProlongSolutionTruncation, SolutionTruncations] \tag{7.1.1}$$

Let us first compute all solution truncations modulo x^4 with given initial value y(0) = 0.

```
> SolutionTruncations(F1, iv=0, 4);
```

$$\left\{x - \frac{1}{6}x^3, x + \frac{1}{6}x^3, x + \frac{4\, RootOf(2\_Z^2-1)\, x^{3/2}}{3} + \frac{x^2}{3} - \frac{23\, RootOf(2\_Z^2-1)\, x^{5/2}}{90} - \frac{247\, x^3}{810} - \frac{9137\, RootOf(2\_Z^2-1)\, x^{7/2}}{30240}, x - \frac{4\, RootOf(2\_Z^2+1)\, x^{3/2}}{3} - \frac{x^2}{3} - \frac{23\, RootOf(2\_Z^2+1)\, x^{5/2}}{90} - \frac{247\, x^3}{810} + \frac{9137\, RootOf(2\_Z^2+1)\, x^{7/2}}{30240}\right\} \tag{7.1.2}$$

The generic solution covers all solutions except those corresponding to the initial values given by the last component.

```
> GenericSolutionTruncation(F1, 0);
```

$$\left\{\left[\_CC + RootOf(\_Z^6 - 6\_Z^5 + (3\_CC^2+15)\_Z^4 + (-12\_CC^2-20)\_Z^3 + (3\_CC^4 + 14\_CC^2+15)\_Z^2 + (-6\_CC^4-4\_CC^2-6)\_Z + \_CC^6 + 3\_CC^4 - \_CC^2 + 1)\, x, \left\{0, \frac{4\, RootOf(3\_Z^2-1)}{3}\right\}\right]\right\} \tag{7.1.3}$$

The initial value 4 RootOf(3 _Z^2-1)/3 is not covered yet.

> $SolutionTruncations\left(F1, iv = \frac{4\, RootOf\left(3\, \_Z^2 - 1\right)}{3}\right);$

$$\left\{ \frac{4\, RootOf\left(3\, \_Z^2 - 1\right)}{3} + RootOf\left(27\, \_Z^2 - 54\, \_Z + 91\right) x, \frac{4\, RootOf\left(3\, \_Z^2 - 1\right)}{3} + RootOf\left(27\, \_Z^2 - 54\, \_Z + 19\right) x - \frac{1}{3} \left(2\, RootOf\left(\_Z^2 + 3\, RootOf\left(3\, \_Z^2 - 1\right) RootOf\left(27\, \_Z^2 - 54\, \_Z + 19\right)\right) RootOf\left(3\, \_Z^2 - 1\right) x^{3/2}\right) \right\} \tag{7.2.1}$$

There are no solutions of negative order:

> $SolutionTruncations(F1, iv = \text{infinity});$

$$\varnothing \tag{7.2.2}$$

Let us note that the built-in command '**dsolve**' does not give such nice solutions.

> $dsolve([F1 = 0, y(0) = 0]);$

$$y(x) = RootOf\left(x - \left(\int_0^{\_Z} 1 \Big/ \left(RootOf\left(\_Z^6 - 6\, \_Z^5 + \left(3\, \_a^2 + 15\right) \_Z^4 + \left(-12\, \_a^2 - 20\right) \_Z^3 + \left(3\, \_a^4 + 14\, \_a^2 + 15\right) \_Z^2 + \left(-6\, \_a^4 - 4\, \_a^2 - 6\right) \_Z + \_a^6 + 3\, \_a^4 - \_a^2 + 1\right)\right) \mathrm{d}\_a\right)\right) \tag{7.2.3}$$

FirstOrderSolve

Let us consider another example.
As it can be seen, we can also compute all solutions by one command. If necessary, we can prolong solution truncations.

> $F2 := y'(x)^2 - y(x)^3 - y(x)^2:$

> $algcurves\text{:-}plot\_real\_curve(subs(\{y'(x) = z, y(x) = y\}, F2), y, z);$



> $SolutionTruncations(F2);$

$$\left[\{[\_CC + RootOf(-\_CC^3 - \_CC^2 + \_Z^2)\, x, \{-1, 0\}]\}, \left\{-1, 0, \frac{4}{x^2}, -1 + \frac{x^2}{4}\right\}, \varnothing\right] \quad \textbf{(8.1)}$$

> $ProlongSolutionTruncation\left(F2, \frac{4}{x^2}, 4\right);$

$$\frac{4}{x^2} - \frac{1}{3} + \frac{x^2}{60} - \frac{x^4}{1512} \quad \textbf{(8.2)}$$

Systems of autonomous AODEs of dimension 1

Let us now consider systems of autonomous AODEs such that the implicitly defined algebraic set is of dimension one, i.e. it is the union of space curves and points.

> $System1 := [y(x)\cdot y'(x)\cdot y''(x) + y'(x)^3 - y(x)\cdot y''(x) - y'(x)^2 = 0, y(x)\cdot y'(x) - 1 - y'(x)^2 - y(x)\cdot y''(x) = 0]$:

> $System1G := subs(\{y''(x) = w, y'(x) = z, y(x) = y\}, System1);$

$$System1G := [y\,z\,w + z^3 - y\,w - z^2 = 0,\ -y\,w + y\,z - z^2 - 1 = 0] \quad \textbf{(9.1.1)}$$

> $plots{:-}spacecurve(\{[1 + t, 1, -1 + 2\,t - 2\,t\verb|^|2 + 2\,t\verb|^|3], [1 + t, 1 - t + t\verb|^|2 - t\verb|^|3, -1 + 3\,t - 6\,t\verb|^|2 + 10\,t\verb|^|3]\}, t = -1\,..1);$



In [3] we have shown that for such systems in $\mathbb{Q}$[y,y',...,y^(k)] there exists a **reduced differential equation** H=0, with H ∈ [y,y'], having the same (non-constant) formal Puiseux series solutions as the original system.
The reduction is done via Thomas decomposition, which is covered by the package '**DifferentialThomas**'.

```
> with(DifferentialThomas);
```

[ComplementOfDecomposition, Display, Equations, Inequations, IntersectDecompositions, LinearCombination, NormalForm, PowerSeriesSolution, Ranking, ReducedForm, ThomasDecomposition, Tools] **(9.2.1)**

```
> Ranking([x], [y, z]);
```

ranking **(9.2.2)**

```
> Thomas1 := ThomasDecomposition(System1);
```

Thomas1 := [DifferentialSystem] **(9.2.3)**

```
> Display(Thomas1);
```

$$\left[\left[y(x)\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right)-1=0,\ y(x)\neq 0\right]\right] \quad \textbf{(9.2.4)}$$

The reduced differential equation can be solved as before.
Additionally, there is a solution expanded around infinity (note the arbitrary constant) and the solution expanded around zero is algebraic. The represention by its minimal polynomial can be computed by our package.

> $SolutionTruncations(y \cdot y' - 1);$

$$\left[\left\{\left[\_CC + \frac{x}{\_CC}, \{0\}\right]\right\}, \left\{RootOf(\_Z^2 - 2)\sqrt{x}\right\}, \left\{2\, RootOf(2\_Z^2 - 1)\sqrt{x} - \frac{2\_CC}{\sqrt{x}}\right\}\right] \quad \textbf{(9.3.1)}$$

> $AlgebraicSolution(y \cdot y' - 1);$

$$\{y^2 - 2x - 1\} \quad \textbf{(9.3.2)}$$

>